-


	Der Ressource-Manager-Controller und die Transaktionsverwaltung in COMCodebook

Yair Alan Griver

Übersetzt von Mathias Gronau


Einführung

Der Ressource-Manager-Controller (RMC) des COMCodebook beinhaltet die grundlegenden Dienste für die Verwaltung der Transaktionen des MTS, die sich über mehrere Komponenten erstrecken. In diesem Abschnitt behandeln wir die Aufgaben und die Implementierung des RMC und seinen Platz innerhalb eines verteilten Systems. Anschließend beschreiben wir die unterschiedlichen Klassen, die in einem VFP-Frontend benötigt werden. Wir beschließen den Abschnitt mit einer Beschreibung der Vorgänge, die während eines Speichervorgangs auftreten.

Der Ressource-Manager-Controller

Wenn unter dem MTS eine bestimmte Funktionalität, beispielsweise ein Save, aufgerufen wird tritt häufig der Fall ein, dass sich dieser Prozess über mehrere Ressourcen erstreckt. Beispielsweise greift die Speicherung einer neuen Rechnung auf die Ressourcen Kunde, Rechnung, Rechnungsdetail und Lagerbestand zu. Um diese Transaktionen über mehrere Komponenten hinweg zu verwirklichen benötigen wir eine Komponente, die mit „Benötigt neue Transaktion“ markiert ist. Die anderen untergeordneten Komponenten müssen entweder als „unterstützt Transaktion“ oder „erfordert Transaktion“ markiert sein. Diese untergeordneten Komponenten werden dann innerhalb der bereits gestarteten Haupttransaktion ausgeführt. Die Haupttransaktion wird vom RMC verwaltet. Wird von der Benutzeroberfläche ein Speichern aufgerufen, bündelt sie sämtliche erforderlichen Ressourcen und übergibt sie an den RMC, der seinerseits eine Transaktion eröffnet und die Verarbeitung an die unterschiedlichen Ressource-Manager delegiert.

Die Entwicklung eines VFP-Frontend mit dem COMCodebook

Bei der Entwicklung des COMCodebook war es uns wichtig, dem Entwickler die Möglichkeit zu eröffnen, so komplexe und funktionale Oberflächen zu entwickeln wie mit früheren Versionen des Codebook. Das bedeutet, dass der Entwickler beliebig viele Relationen in unterschiedlichen Konfigurationen erstellen kann. Auf diese Weise werden die Datenquellen und Ressource-Manager nur ein Mal erstellt, während die Relationen zwischen den Ressourcen zur Laufzeit durch die Anwendung erstellt werden. Nachdem die Komponenten der mittleren Schicht für das Unternehmen ein Mal erstellt sind, besteht das Hinzufügen eines neuen Frontend lediglich aus der Definition der Relationen zwischen den Komponenten und dem Entwickeln einer entsprechenden Benutzeroberfläche.

Diese Bedingung wird in einem VFP-Frontend mit dem Ressource-Manager-Proxy (RMP), dem Ressource-Manager-Proxy-Loader (RMPL) und dem Ressource-Manager-Proxy-Controller (RMPC) erfüllt.

Der Ressource-Manager-Proxy

Der RMP repräsentiert die Ressource-Manager-Komponente gegenüber der Benutzeroberfläche. Er gehört zur mittleren Schicht. Seine Aufgabe ist es, sämtliche Ressourcen der mittleren Schicht in ein der Benutzeroberfläche verständliches Format zu übersetzen. So übersetzt er z. B. ein ADO-Recordset in einen VFP-Cursor, so dass der Entwickler der Benutzeroberfläche in seiner gewohnten Art arbeiten kann.

Im RMP können auch die Relationen, Fehlermeldungen, die Primärschlüssel der übergeordneten und die Fremdschlüssel der untergeordneten Ressourcen usw. abgelegt werden. Im Wesentlichen hat der RMP viele Aufgaben des Business-Objekts des alten Codebook übernommen.

Der Ressource-Manager-Proxy-Loader

Der RMPL ist einfach ein Lademechanismus, der es einem Entwickler ermöglicht, RMPs visuell auf ein Formular zu ziehen, so dass sie zur Laufzeit des Formulars automatisch instanziiert werden. Mit Hilfe der Eigenschaften von cResourceManagerProxyLoader, die sich in CUTILS.VCX befindet, kann der Entwickler angeben, welche RMPs geladen und welche Relationen erstellt werden sollen.

Der Ressource-Manager-Proxy-Controller

Der RMPC verfolgt die RMPs eines Formulars. Wenn ein RMP erstellt wird, registriert er sich beim RMPC. Nachdem der letzte RMP geladen wurde, ist der RMPC für die Synchronisation untergeordneter RMPs mit ihren übergeordneten RMPs zuständig. Der RMPC ist das direkte Äquivalent zum RMC der mittleren Schicht und kommuniziert auch mit ihm.

Schritt für Schritt durch ein Formular

Die Initialisierung des Formulars

Wenn ein Formular instanziiert wird, wird der Ressource-Manager-Proxy-Controller im Load() des Formulars erstellt. 

*---------------- Location Section ---------------------

*} Library: CFORMS.VCX

*} Class: CResourceForm

*} Method: Load()

*-------------------------------------------------------

*) Description: 

*) Creates the Resource Manager Conrol Object

*-------------------------------------------------------

DODEFAULT()

THIS.CreateResourceManagerControlObject()

Der folgende Code zeigt, wie das Framework erkennt, welches Objekt das Formular als seinen Ressource-Manager-Proxy-Controller benutzen soll. Es nimmt den in der Eigenschaft cResourceManagerControlClass gespeicherten Namen der Klassendefinition. Der Vorgabewert für diese Eigenschaft ist cResourceManagerProxyCollection, der in der Klassenbibliothek CCOLLECT gespeichert ist (im I-Layer benutzt er iResourceManagerProxyCollection). Das sich daraus ergebende Objekt speichert das Formular in seiner Eigenschaft .oRMControl).

*---------------- Location Section ---------------------

*} Library: CFORMS.VCX

*} Class: cBaseForm

*} Method: CreateResourcemanagerControlObject()

*-------------------------------------------------------

*) Description:

*) Creates the object that tracks and controls communication

*) between the many Resource Mananger Proxys that are

*) present on the form.

*-------------------------------------------------------

WITH THIS

IF !EMPTY( .cResourceManagerControlClass )

.oRMControl = CREATEOBJECT( .cResourceManagerControlClass, THIS )

ENDIF

ENDWITH

RETURN .T.

Während des Ladevorgangs des Formulars ist es auch für die in ihm enthaltenen Objekte an der Zeit, den Ladevorgang einzuleiten. Zu diesen Objekten gehören auch die Ressource-Manager-Proxy-Loader. Die Klassendefinition dieser Objekte finden Sie in CUTILS.VCX unter dem Namen CResourceManagerProxyLoader. Wenn Sie weitere Informationen wünschen können Sie jetzt dort die Methode Init() ansehen. Dort werden Sie erkennen, dass es die wichtigste Funktion des RMProxyLoader ist, die Objekte des Ressource-Manager-Proxy zu erstellen und sie im RMController des Formulars zu registrieren. Dies geschieht durch die Übergabe einer Referenz auf den neu erstellten Ressource-Manager-Proxy-Objekt an die Methode .Add() des Controllers.

Dem RMPC des Formulars einen Ressource-Proxy-Manager hinzufügen

Die Objekte des Ressource-Manager-Controllers enthalten eine Referenz auf alle Ressource-Manager-Proxy-Objekte des Formulars. Die Methode Add() ist die zentrale Möglichkeit für die Registrierung der Proxy-Objekte. Diese Methode führt vier grundlegende Aufgaben aus. Als Erstes wird geprüft, ob der RM-Proxy bereits hinzugefügt wurde. Ist dies nicht der Fall wird als Zweites eine neue Eigenschaft als Teil der Collection angelegt und es wird darin eine Referenz auf den Proxy gespeichert. Als dritte Aufgabe wird dem Array .aChildren ein neuer Datensatz hinzugefügt, der den neu hinzugefügten Proxy speichert. Letztendlich wird der Zähler der Controls um eins hochgezählt.

Über- und untergeordnete Manager-Proxy-Objekte verbinden

Nachdem alle Ressource-Manager-Objekte erstellt und mit dem Formular registriert wurden, wird die Verarbeitung mit der Methode Init() fortgeführt, die dem .oRMController mitteilt, dass es an der Zeit ist, die untergeordneten Ressourcen mit ihren jeweiligen übergeordneten zu verbinden.

*---------------- Location Section ---------------------

*} Library: CFORMS.VCX

*} Class: CResourceForm

*} Method: Init()

*-------------------------------------------------------

*) Description:

*-------------------------------------------------------

LOCAL llRetVal

llRetVal = DODEFAULT()

IF llRetVal

THIS.oRMControl.LinkParentChildResources()

ENDIF

RETURN llRetVal

In der Methode LinkParentChildResource() scannt die Ressource-Proxy-Collection jeden registrierten RMP und führt zwei wichtige Aufgaben aus. Als Erstes speichert sie den zu dem entsprechenden RMP gehörenden Ressource-Manager der mittleren Schicht in der zweiten Spalte des Array. Als Zweites wird geprüft, ob zu dem aktuellen Proxy-Objekt ein übergeordnetes Objekt existiert. Ist dies der Fall, registriert sie das untergeordnete Objekt mit einem Aufruf der Methode .RegisterChildObject des übergeordneten Objekts.

Beachten Sie, dass der RMPC jetzt alle RMPs des Formulars kennt und auch Informationen darüber hat, für welche Komponente der mittleren Schicht der RMP den Proxy bildet.

Das Speichern von Informationen

Jetzt verfügen wir über einen RMPC mit allen von uns benötigten Informationen. Lassen Sie uns ergründen, was passiert wenn der Anwender auf den Button Speichern klickt.

Als Erstes erhält das Formular über seine Methode Save() die Nachricht und gibt sie an seinen Proxy-Controller weiter.

*---------------- Location Section ---------------------

*} Library: CFORMS.VCX

*} Class: CResourceForm

*} Method: Save()

*-------------------------------------------------------

*) Description:

*) This method is responsible for the following things:

*) a. Locking the form during the saving processing

*) b. Unlocking the form after the save processing

*) is complete.

*) c. Passing the save message through to the 

*) Resource Manager Proxy Controller.

*) d. Informing the application that a Save

*) type of activity took place.

*-------------------------------------------------------

LOCAL lnRetVal, llOldLockScreen

=LockScreen( .T., @llOldLockScreen )

lnRetVal = THIS.oRMControl.Save()

THISFORM.Refresh()

=LockScreen( llOldLockScreen )

goApp.RecordActivity( "FormSave" )

RETURN lnRetVal

Jetzt übernimmt die Methode Save() des Ressource-Manager-Controllers und führt den folgenden Code aus:

*---------------- Location Section ---------------------

*} Library: CCOLLECT.VCX

*} Class: CResourceManagerProxyCollection

*} Method: Save()

*-------------------------------------------------------

*) Description:

*) Template method that controls the saving the information

*) in the user interface to the data source on the 

*-------------------------------------------------------

LOCAL lnRetVal, loRMController, lcXML

lnRetVal = FILE_CANCEL

IF THIS.oPrimaryResource.AllowSave()

IF THIS.oPrimaryResource.WriteBuffer()

IF .NOT. THIS.oPrimaryResource.IsAdding()

IF .NOT. THIS.oPrimaryResource.IsChanged()

RETURN FILE_CANCEL

ENDIF

ENDIF

 

DO CASE

CASE .NOT. THIS.CreateRMController( @loRMController )

CASE .NOT. THIS.GetEnvDefinition( @lcXML )

CASE .NOT. THIS.SetUpEnvironment( loRMController, lcXML )

CASE .NOT. ( THIS.MarshallData( loRMController ) == FILE_OK )

OTHERWISE

THIS.oPrimaryResource.BeforeSave()

THIS.oPrimaryResource.oControlContainer.BeforeSave()

lnRetVal = loRMController.Save()

THIS.oPrimaryResource.oControlContainer.AfterSave( lnRetVal )

THIS.oPrimaryResource.oControlContainer.RefreshControls( THIS.oPrimaryResource )

THIS.oPrimaryResource.AfterSave( lnRetVal )

ENDCASE

ENDIF

ENDIF

IF lnRetVal = FILE_OK

goApp.DisplayMessage( THIS.oPrimaryResource.cSaveSuccessMessageKey )

ELSE

goApp.DisplayMessage( THIS.oPrimaryResource.cSaveFailureMessageKey )

ENDIF

RETURN lnRetVal

Diesen Code möchte ich Ihnen kurz erläutern. Sehen wir uns eine Template-Methode an. Eine Template-Methode führt genau eine Aufgabe aus. Sie definiert, was in welcher Reihenfolge ausgeführt wird. Sie definiert nicht, wie diese Aufgabe ausgeführt wird. Wie etwas getan wird steht in anderen Methoden, die bei Bedarf einfach überschrieben werden können. Auch ein relativ unerfahrener Leser sollte in recht kurzer Zeit die Funktionalität einer Template-Methode einfach durch das Lesen der Methodenaufrufe erkennen können. Der Vorteil für den Entwickler ist die Möglichkeit, die Verarbeitung durch ein einfaches Überschreiben der Methode in einer abgeleiteten Klasse zu verändern, ohne dabei die Template-Methode modifizieren zu müssen. Wenn Ihnen z. B. unser Vorgehen beim Erstellen des RM-Controllers nicht zusagt, können Sie durch einfaches Überschreiben der Methode .CreateRMController() im I-Layer Ihren eigenen Controller erstellen.

Zusammenfassend gesagt fragt diese Methode alle RMPs, ob Informationen des Frontend gespeichert werden sollen. Ist dies der Fall kommen wir zum zentralen Punkt der Methode Save().

Als erster Schritt wird eine Referenz auf das Ressource-Manager-Objekt in der mittleren Schicht (die Methode .CreateRMController()) erstellt. Als Nächstes erhalten wir eine Definition der Umgebung, die der Entwickler der Benutzerschnittstelle erstellt hat. Dabei handelt es sich um eine Liste aller Ressource-Manager-Proxies, den Namen der ihnen übergeordneten Objekte (soweit vorhanden) sowie die Information, welches Objekt der primäre Ressource-Manager für diese Konfiguration ist. Anschließend erhält der Ressource-Manager-Controller der mittleren Schicht die Anweisung, diese Umgebung mit Hilfe der Ressource-Manager-Objekte der mittleren Schicht statt mit den Ressource-Manager-Proxy-Objekten der Benutzeroberfläche nachzubilden. Schließlich werden die Daten aus der Benutzeroberfläche in die entsprechenden Objekte der mittleren Schicht übergeben.

Sind alle diese Aufgaben erfolgreich beendet worden, ruft der Ressource-Manager-Proxy-Controller die Methode Save() des Ressource-Manager-Controllers auf.

* CAbstractRMController in CResourceManagers.PRG 

*--------------

FUNCTION Save()

*--------------

LOCAL lnRetVal, loContext

lnRetVal = FILE_CANCEL

 

THIS.CreateMTSContextObject( @loContext )

WITH THIS.oPrimaryResource

IF .Validate()

lnRetVal = .Save()

ENDIF

ENDWITH

THIS.ExecuteMTSSetOperations( loContext, lnRetVal )

RETURN lnRetVal

ENDFUNC

Beachten Sie, dass diese Methode der mittleren Schicht abschließend durch das Erstellen eines MTS-Kontext-Objekts eine MTS-Transaktion ermöglicht. Anschließend wird die primäre Ressource angewiesen, sich selbst zu prüfen. Nachdem die primäre Ressource mit allen untergeordneten Objekten die Validierung durchlaufen hat, wird die Methode Save() ausgeführt. Das Ergebnis dieses Methodenaufrufs wird der Methode ExecuteMTSSetOperatons() übergeben, wo die Transaktion entweder abgeschlossen oder zurückgenommen wird. Ein Ablaufdiagramm des Speichervorgangs sehen Sie hier.

Zusammenfassung

Die physische Trennung der Daten in den verschiedenen Schichten, die Notwendigkeit, beim Ändern von Daten die Transaktionen des MTS einzusetzen und der Anspruch, dem Entwickler die Möglichkeit zu geben, in der Benutzeroberfläche komplexe Relationen aufbauen zu können, führen uns zu einigen interessanten Lösungen. Zu diesen Lösungen kommen wir durch die Verwendung eines Controller-Objekts sowohl in der Benutzeroberfläche als auch in der mittleren Schicht. Der Entwickler definiert seine Ressource-Manager-Proxy-Objekte in der Schicht der Benutzeroberfläche. Diese Umgebung wird in der mittleren Schicht automatisch nachgebildet. Die Daten werden der mittleren Schicht übergeben, wo sie innerhalb einer Transaktion des MTS verarbeitet, geprüft und gespeichert werden.

